	Sub-Classing von PageFrames, um schnell auf jede Seite zuzugreifen

Lisa Slater Nicholls

Am Anfang schien alles ganz einfach: Wie kann ich PageFrames Funktionalität hinzufügen, damit auf jede Seite sofort zugegriffen werden kann? Nun werden Sie wahrscheinlich denken: „Ich kann doch auf jede Seite zugreifen. Ein Klick auf den Tab der Seite genügt. Schon wechselt die aktive Seite des PageFrames und steht im Vordergrund. Wenn wir noch einen Hotkey hinzufügen, können wir auch mittels der Tastatur auf die Seiten zugreifen.“

Die natürlichen Fähigkeiten der PageFrame-Basisklasse von VFP bieten eine ganze Reihe Möglichkeiten, auf die Seiten zuzugreifen. Aber was passiert, wenn Ihr PageFrame gar keine Tabs hat?

Sobald Sie ein PageFrame-Tab-Property auf .F. setzen, hat der Anwender mit der Maus keinen Zugriff mehr auf diese Seite. Interessanterweise ist der Zugriff über die Tastatur weiterhin möglich. Dafür muß der Anwender nur den Hotkey kennen.


Die Dialog-Box ohne Tabs

Abbildung 1
Die Assistenten von Visual FoxPro sind Beispiele für PageFrames ohne Tabs

Weshalb sollten Sie einen PageFrame ohne Tabs erstellen? Denken Sie nur einmal an die Assistenten von VFP. Jeder ist eine lineare Abfolge von Schritten, bei der der Anwender an die Hand genommen und durch eine Reihe von Optionen geführt wird, bis er am Ende sein gewünschtes Ergebnis erhält.

Jeder Assistent enthält eine Gruppe von Buttons, die dem Benutzer erlauben, den gesamten Vorgang abzubrechen, einen Schritt vor oder zurück zu blättern oder auch die vorgegebenen Optionen zu akzeptieren und so den Prozeß zu beenden. Abbildung 1 zeigt ein Beispiel  aus dem Berichtsassistenten für Gruppen/Summen. Die Inhalte jedes Schrittes sind auf einer einzelnen Seite, abhängig von den Einstellungen, die der Anwender vorher getroffen hat (der Button Fertigstellen beendet die Auswahlseiten und stellt den Bericht unter Verwendung der Vorgabewerte fertig).

Wie VFP wird auch Ihre Anwendung Auswahlblöcke enthalten, die aufeinander aufbauen und in einer bestimmten Reihenfolge ausgeführt werden sollten. In den Zeiten vor Visual FoxPro haben Sie diese Blöcke wahrscheinlich als eine Serie von zusammengehörenden Dialogboxen dargestellt, um die richtigen Inhalte und Reihenfolgen sicherzustellen.

PageFrames sind ein effizienter Weg, um Sequenzen wie die Assistenten anzuzeigen. Dialogboxen mit Tabs sind aber keine angemessene Benutzerschnittstelle für Assistenten. Die Tabs ermutigen die Anwender, die nötigen Schritte in einer beliebigen Reihenfolge durchzuführen und den PageFrame von jeder beliebigen Seite aus zu beenden, ohne daß die Sequenz logisch abgeschlossen ist. Denken Sie zum Beispiel an die Dialogbox Optionen in Visual FoxPro. Würde es irgendeinen Sinn machen, die Projekte-Optionen vor oder nach den Allgemein-Optionen wählen zu müssen? Natürlich nicht, die Reihenfolge, in der Sie die Optionen bearbeiten, ist gleichgültig, da die Optionen auf jeder Seite unabhängig von den anderen Seiten sind (Diese Dialogbox ist ein besonders gutes Beispiel für dieses Problem, da sie die Tabs in mehreren Reihen übereinander zeigt. Wenn Sie Ihre eigenen Dialogboxen mit Tabs entwickeln, setzen Sie die TabStretch-Eigenschaft auf 0, um die Tabs in einer einzelnen Reihe anzuzeigen.)

Ich suchte eine einfache Methode, auf PageFrames zuzugreifen, ohne dafür Tabs einzusetzen. Die Anwenderschnittstelle sollte die Erfordernisse der Assistenten und anderer Dialogboxen ohne Tabs, die in einer linearen Sequenz abgearbeitet werden,  umfassen, und meine neue Zugriffsmethode sollte den Anwendern diese Reihenfolge nahebringen.

Die Assistenten verfügen für diesen Zweck in der rechten oberen Ecke über ein Kombinationsfeld. Die Liste des Feldes zeigt eine Aufstellung der Schritte (bzw. Seiten), aber diese Lösung ist recht plump und müllt den Bildschirm ziemlich voll. Statt dessen habe ich mich für eine eigene Methode entschieden, die immer von jedem PageFrame aufgerufen werden kann, ohne daß der Anwender explizit diese Kontrollelemente aufrufen muß.

Die Version, die ich in meiner Klassenbibliothek erstellt habe, bindet die Methode an einen rechten Mausklick auf den PageFrame, aber Sie werden leicht erkennen, wie sich diese Funktionalität auf einen Tastendruck oder andere Aktionen des Anwenders legen läßt.

Die PageFrames von Visual FoxPro haben viele gute Eigenschaften, aber Einfachheit gehört nicht dazu, wie dieser kurze Überblick über ihr Benutzerschnittstellendesign zeigt. Auf dem Weg zur Lösung meines „einfachen Problems“ behandeln wir:

· Container und Collection-Eigenschaften

· den Gebrauch eines Standardbefehls von FoxPro, DEFINE POPUP, zusammen mit OOP-orientierten Kontrollelementen

· Bildschirm- und mauskoordinierte Referenzierung

· spezielle Builder

· die programmatische wie auch die visuelle Erstellung von Klassen

· einige interessante Bugs

Willkommen zu VFPs OOP-Theorie und -Praxis, wo jedes Problem organisatorisch mit jedem anderen verbunden ist..

Grundlegender Code zur Manipulation von Seiten und Popups

Die RightClickMethode meiner PageFrame-Klasse ruft eine Liste zur Auswahl der Seiten auf, die direkt unter der Maus angezeigt wird. Den Code dazu sehen Sie in Listing A.

* Listing A
*

IF THIS.Visible AND THIS.PageCount > 0

LOCAL lnIndex, lcPrompt, lcFont, lcStyle, lcPopup

lcPopup = SYS(2015)

DEFINE POPUP (lcPopup) IN SCREEN

FOR lnIndex = 1 TO THIS.PageCount

lcPrompt = IIF(THIS.Pages[lnIndex].Enabled, ;





"","\") + THIS.Pages[lnIndex].Caption

lcStyle =IIF(THIS.Pages[lnIndex].FontItalic, „I", „") + ;




IIF(THIS.Pages[lnIndex].FontOutline, „0", „") + ;




IIF(THIS.Pages[lnIndex].FontShadow, „S", „") + ;




IIF(THIS.Pages[lnIndex].FontStrikethru, „-„, „") + ;




IIF(THIS.Pages[lnIndex].FontUnderline, „U", „")

lcFont = „ ‘"+THIS.Pages[lnIndex].FontName+"’, „+ ;




ALLT(STR(THIS.Pages[lnIndex].FontSize))+ ;




„ -STYLE „ + ;




IIF(EMPTY(lcStyle),"’N’", „’"+lcStyle+"’")

DEFINE BAR THIS.Pages[lnIndex].PageOrder ;




OF (lcPopup) PROMPT lcPrompt ;




FONT &lcFont

ENDFOR

ON SELECTION POPUP (lcPopup) DEACTIVATE POPUP

ACTIVATE POPUP (lcPopup) AT MROW(„"), MCOL(„") ;





BAR THIS.ActivePage

IF NOT EMPTY(BAR())



THIS.ActivePage = BAR()
ENDIF

RELEASE POPUP(lcPopup)

ENDIF

Die Prozedur definiert das erforderliche Popup. Anschließend wird es unter Verwendung der ActivePage-Eigenschaft des PageFrames aktiviert, um die aktuelle Seite im Popup zu markieren. Wenn der Anwender das Popup schließt, speichert die Prozedur einen neuen Wert in der ActivePage-Eigenschaft des Pageframes, um die Seite, die der Anwender gewählt hat, in den Vordergrund zu holen. 

Die Definition des Kombinationsfeldes benutzt mehrere Features von VFP, die Beachtung verdienen. Die FOR/ENDFOR-Schleife, die die Einträge definiert, benutzt eine Array-Eigenschaft der PageFrames: Pages[]. Der PageFrame ist ein Containerobjekt mit eingebetteten Objekten, den Seiten. Die Pages[]-Eigenschaft enthält Referenzen zu allen Seiten mit der Anzahl der Seiten, die in der PageCount-Eigenschaft gespeichert sind.

Alle Containerobjekte haben in der VFP-Basisklasse ähnliche Array-Eigenschaften. Zum Beispiel hat ein Formular eine Controls[]-Collection und einen entsprechenden ControlCount.

Wir nennen diese Eigenschaften Collections, da dieser Begriff den Inhalt der Eigenschaften exakt beschreibt. Außerdem unterscheidet der Begriff diese Eigenschaften von normalen VFP-Arrays und Array-Objekten, die Sie Ihren Objekten hinzufügen. Die Unterscheidung ist wichtig, da Sie die normalen Array-Funktionen wie ASCAN()in Collection-Eigenschaften nicht benutzen können.

Wenn Sie sich meine FOR/ENDFOR-Schleife ansehen, könnten Sie auf den Gedanken kommen, daß ich die Einträge in der korrekten Reihenfolge für das Menü zuweise, da die Schleife die Pages[]-Collection des PageFrames der Reihe nach abarbeitet. Wie auch immer, die Pages[]-Collection enthält die Referenzen der Seiten, in der die Seiten definiert wurden, nicht in der Reihenfolge, in der sie erscheinen. Aus diesem Grunde benutze ich „Pages[expN].PageOrder“ um die Eintrags-Nummer für meine Definition zu erhalten.

Sie sollten feststellen, daß ich die RELATIVE-Klausel im Kommando DEFINE POPUP nicht benutzt habe. Normalerweise benutzen Sie die DEFINE BAR-Kommandos in der Reihenfolge, in der sie im Menü erscheinen sollen, und die RELATIVE-Klausel stellt diese Reihenfolge noch einmal sicher. Im SYSMENU erlaubt die RELATIVE-Klausel es den Optionen des System-Menüs, zum Beispiel _med_undo (der Rückgängig-Option oben im Bearbeiten-Popup), vor Ihren eigenen Maskenelementen zu erscheinen, obgleich dem System-Menü durch das System sehr hohe Eintrags-Nummern zugewiesen werden.

Wenn die zweite Seite definiert wurde, ist sie zufällig der 5. Eintrag in PageOrder. DEFINE BAR THIS.Pages[2].PageOrder gibt mir die „5“ als Eintrags-Nummer zurück. Ohne die RELATIVE-Klausel, sogar obwohl ich nur die zweite BAR definiert hatte, bekommt der Eintrag ein absolutes BAR 5. Im Popup wird Platz für Einträge mit geringeren Nummern vor dem Eintrag reserviert. 

Durch diesen Trick bekommt der Anwender das Popup in der Reihenfolge angezeigt, die der Entwickler für die Seiten vorgesehen hat. Genauso wichtig ist es, daß der vom Anwender gewählte Bar() immer mit der PageOrder des Pageframes übereinstimmt, so daß man BAR() der ActivePage-Eigenschaft des PageFrames zuordnen und zu der richtigen Seite springen kann, ohne die Nummer des Bar() ändern zu müssen. Dementsprechend kann die BAR-Klausel des ACTIVE POPUP-Statement genutzt werden, um festzulegen, welche Popup-Option hervorzuheben ist. 

Ich benutze die Caption-Eigenschaft jeder Seite, um den Popup-Prompt für jede Seite zu bestimmen, der auch als Beschreibung jeder Seite (oder jedes „Schrittes“) dient. Sie können die Überschriften für die Tabs sogar einer Dialogbox ohne Tabs zuordnen. Tatsächlich ist es eine gute Lösung, Überschriften zuzuordnen, um die Möglichkeit zu schaffen, Tastenkombinationen zu erstellen, die dem Right-Click Page-Zugriff entsprechen.

Aktuell nicht aktivierbare Seiten sind im Popup dunkel unterlegt (und können nicht ausgewählt werden), da ich den Überschriften bei der Definition einen Backslash vorangestellt habe. Ich benutze auch die Font-Eigenschaft der Seite, um die Schriftart für jedes Popup festzulegen. Dazu bietet sich die neue „DEFINE BAR ... FONT“-Klausel von VFP an.

Sie werden feststellen, daß ich MROW(““) und MCOL(““) benutze, um die Koordinaten des Popup relativ zum Screen, nicht bezogen auf das aktive Formular, zu definieren. Da MROW(), MCOL() und die AT-Klausel von ACTIVATE POPUP alle mit Foxeln arbeiten, werden diese Koordinaten immer übereinstimmen.

Ich habe für ACTIVATE POPUP AT anstatt der Mauskoordinaten DEFINE POPUP FROM gewählt, da die ACTIVATE-Syntax das Popup verläßlich auf dem Bildschirm plaziert, nicht in das aktive Fenster. Ich hätte eine Syntax wie diese benutzen können:

DEFINE POPUP (lcPopup) ;


FROM MROW(_SCREEN.ActiveForm.Name), ;


MCOL(_SCREEN.ActiveForm.Name) ;


IN WINDOW _SCREEN.ActiveForm.Name

Es tritt jedoch kein Fehler auf, wenn das Popup auf dem Screen plaziert wird, da das Popup immer ganz oben im aktuellen Fenster erscheint. Das Plazieren des Popup scheint auch exakter zu sein, wenn man es mit dem ACTIVATE POPUP-Befehl spezifiziert.

Funktioniert das auch?

Fügen Sie meinen Code der RightClick-Methode Ihrer PageFrame-Klasse hinzu und setzen Sie eine Instanz dieser Klasse in ein Probeformular, um es zu testen. (Behalten Sie die Tabs jetzt noch.) Sie werden feststellen, daß der RightClick das Popup problemlos aufruft und dieser auch fehlerfrei eine Seite selektiert – solange man die Tabs richtig anklickt, aber nirgendwo anders innerhalb des PageFrame! Was ist falsch gelaufen?

Erinnern Sie sich, daß der PageFrame ein Container-Steuerelement ist und jede Seite, die es enthält, ein separates Objekt ist. Obwohl die Tabs ein Teil des PageFrame-Objektes zu sein scheinen, erscheint das Ereignis in der Seite und nicht im PageFrame, wenn man einen anderen Teil als die Tabs auf dem PageFrame anklickt.

Eingebettete Kontrollelemente sind von ihren Klassen abgeleitet, nicht von ihren Containern. 

Zum Beispiel erhält ein Label-Kontrollelement seine BackColor-Eigenschaft von der Klasse, von der es abgeleitet ist, nicht von der BackColor-Eigenschaft des Formulars, auf das es zufällig gesetzt wurde. Wenn Sie Code in die Click-Methode des Formulars eingeben, wird diese Methode nicht ausgeführt, wenn der Anwender den Label anklickt, der sein eigenes Click-Ereignis besitzt.

In diesem Fall haben die Seiten ihr eigenes RightClick-Ereignis, und daher haben wir den Seiten bislang noch nicht mitgeteilt, welcher Code ausgeführt werden soll, wenn ein RightClick auftritt.

Natürlich wollen wir den Code nicht in jeder Methode wiederholen. Anstelle dessen setzen Sie die folgende Syntax in jede RightClick-Methode der Seiten:


THIS.Parent.RightClick()

Dieser Code delegiert von jeder Seite an den PageFrame und führt dessen RightClick-Code aus. In der objektorientierten Programmierung beschreibt das Wort delegieren die Anweisung eines Objektes an ein anderes, eine Aufgabe auszuführen. Das geschieht im Prinzip auf die gleiche Weise, wie Sie eine Aufgabe an einen Mitarbeiter delegieren, der die richtige Ausrüstung und die richtigen Kenntnisse für die Ausführung dieser Arbeit besitzt.

Die Delegierung hängt nicht von den Containern und den von ihnen abgeleiteten Klassen ab, und genausowenig muß die Aufgabe notwendigerweise auf einer Methode mit demselben Namen beruhen. Zum Beispiel kann eine Befehlsschaltfläche in ihrer Click-Methode folgenden Code enthalten, mit dem an eine vollständig getrennte Einheit delegiert und dort die notwendige Aktion ausgeführt wird:

...(falls notwendig, spezieller Code der Befehlsschaltfläche)

Delegierung an die DoubleClick-Methode eines getrennten Formulars:

MyOtherForm.DoubleClick()

Sie können auch benutzerdefinierte Methoden delegieren:

THISFORM.MyMethod()

Führen Sie das Testformular noch einmal aus. Dieses Mal wird sie funktionieren, unabhängig davon, an welcher Stelle der Seite Sie die rechte Maustaste drücken. Ihr Formular sollte in etwa so aussehen wie Abbildung 2. (Nun können Sie die Tabs entfernen, indem Sie die Syntax 

TestForm.MyPageFrame1.Tabs = .F.

Im Befehlsfenster eingeben.) In Abbildung 2 haben wir die Reihenfolge der Seitendefinitionen in den Überschriften der Seiten 2, 3 und 4 erhalten, aber Sie werden feststellen, daß die Reihenfolge der Einträge des Kombinationsfeldes, sondern stattdessen die Reihenfolge der Seiten wiedergibt.

 Abbildung  2 
Eine Instanz meiner MyFrame-Klasse zeigt das Popup, das die PageOrder-, Enabled-, Caption- und Font-Eigenschaften jeder Seite wiederspiegelt
Nicht gut genug: Öffnen Sie den Builder

Möchten Sie Zugang zu der RightClick-Methode jeder Seite haben, um diesen Code zu ergänzen? Ich nicht. Wir haben nicht die Möglichkeit, benutzerdefinierte Versionen des Seiten-Objektss im Klassendesigner zu erstellen. Wenn wir es versuchen, könnten wir nur ein benutzerdefiniertes Seiten-Objekt mit der entsprechenden RightClick-Methode erstellen und Instanzen dieser Klasse in den PageFrame integrieren. Vielleicht wäre es aber auch möglich, daß das Eigenschaften-Fenster uns erlaubt, ein benutzerdefiniertes Seiten-Objekt auszuwählen, wenn wir zuvor die PageCount-Eigenschaft des PageFrame entsprechend anpassen. 

Da VFP 3.0 diese Fähigkeit nicht besitzt, können Sie mit Hilfe einer Tastenkombination den Code in die Seiten Ihrer benutzerdefinierten PageFrame-Klasse eingeben, indem Sie einen Builder benutzen. Den dafür benötigten Code sehen Sie in Listing B. 

* P_BUILD.PRG

* Spezieller Builder für MyPageFrame Class

   PARAMETERS toRef, t2, t3

   LOCAL lnindex, lopage

   * Hier wird der Builder im PageCount angepaßt

   DO Form P_BUILD TO toRef.PageCount

   FOR lnindex = 1 TO toRef.PageCount

      lopage = EVAL("toRef.Page"+ALLTR(STR(lnindex)))

      loPage.WriteMethod("RightClick",;

                  "THIS.Parent.RightClick()")

   ENDFOR 

RETURN

Sie können diesen Code erweitern, indem Sie ein modales Formular im Builder aufrufen, das abfragt, wieviele Seiten der PageFrame haben soll, und PageCount entsprechend anpassen. Ich habe das Beispiel P_BUILD.SCX von der FoxPro Developer’s Resource Disk beigefügt. (Benutzen Sie die UnLoad-Methode, um zu erreichen, daß das modales Formular einen Wert zurückgibt.) Ich bevorzuge es, den Wert des modalen Formulars direkt in die PageCount-Eigenschaft des PageFrame zurückzugeben, wie ich es in der auskommentierten Zeile in Listing B dargestellt habe. Dadurch sind weder eine zusätzliche Variable noch weitere Schritte von meiner Seite notwendig.

Wie können Sie sicherstellen, daß der Code dieses Builders nur für Ihre spezielle PageFrame-Klasse aktiviert wird? Es gibt mehrere Möglichkeiten, einen Builder mit Klassen zu verbinden. Allerdings hindert ein Bug im ausgelieferten Produkt die meisten Builder daran, mit PageFrames zusammen-zuarbeiten. (Dieser Bug könnte in der BUILDER.APP stecken; es ist jedoch wahrscheinlicher, daß er sich direkt in VFP befindet. Das würde erklären, warum das Produkt ohne einen PageFrame-Builder ausgeliefert wird. Sie werden feststellen, daß das Eintragen eines PageFrame-Builders nicht so funktioniert, wie Sie es erwarten. Dies ist auch der Fall, wenn Sie eine Builder-Eigenschaft für ein Objekt einer PageFrame-Klasse nutzen.)

Im Moment ist es am besten, einen Wrapper um den VFP BUILDER.APP oder jeden anderen Builder zu legen, den Sie benutzen. Initialisieren Sie die Systemvariable _BUILDER mit dem Namen Ihres Wrappers und verwenden Sie den Code aus Listing C.

* BWRAPPER.PRG

* Geht in die _BUILDER Systemvariable,

* um Ihnen das Erstellen eines Builders

* für eine PageFrame-Klasse zu ermöglichen...

PARAMETERS toRef, t2, t3

   * Dies könnte auch ein CASE sein, 

   * mit dem Sie auch andere Builder

   * einbinden können...

   IF LOWER(toRef.Class) = "mypageframe"

      DO P_Build WITH toRef, t2, t3

   ELSE

      DO HOME()+"BUILDER.APP" WITH toRef, t2, t3

      * oder ein "Standard"-Builder Ihrer Wahl

   ENDIF

   RETURN

Natürlich könnte dieser Wrapper viele CASEs beinhalten statt der einfachen IF/ENDIF-Abfrage. Jedes CASE kümmert sich um eine andere besondere Situation, für die Sie einen Builder einsetzen, der nicht in Ihrem Standard-Builder eingetragen ist. Stellen Sie lediglich sicher, daß dieselben Parameter, die der Wrapper von VFP empfängt, an jeden Builder in jedem Ausdruck des CASE (einschließlich des OTHERWISE, das den Standard-Builder aufruft) weitergegeben wird.

Sie haben nun eine vernünftige Anpassung der PageFrame-Klasse, die Sie natürlich wie ich in allen Ihren Formularen ersetzen können. Ihr Builder stellt sicher, daß Sie keine unnötige Arbeit verrichten müssen, um den notwendigen Code in jede Ihrer Seiten zu stellen.

Obwohl diese Technik eine große Flexibilität im Hinblick auf die Nutzung von PageFrames aufweist, gibt es trotzdem Grenzen. Zur weiteren Information lesen Sie den Kasten „Eine programmatische Page-Klasse...“

Nicht nur für Dialoge ohne Tabs

Auch Dialogboxen mit Tabs können von dieser Methode des Seitenzugriffs profitieren. Nehmen wir einmal an, sie erzeugen eine Eingabemaske, die aus einer langen Reihe manuell auszufüllender Formulare besteht, zum Beispiel in einer Anwendung auf dem Gebiet der Versicherungen oder für einen Fragebogen bei Meinungsumfragen. Dialogboxen mit Tabs, mit einem Link auf die entsprechende Seite des Handbuchs, sind gut geeignet, Platz auf dem Bildschirm zu sparen und das Aussehen der ursprünglichen Maske gleichzeitig zu reproduzieren.

Ihre Anwender werden es zu schätzen wissen, daß sie in der Lage sind, die Seiten zu wechseln, indem sie mit der rechten Maustaste auf jeden Punkt des PageFrame klicken, anstatt mit der Maus zu den Tabs am oberen Bildschirmrand gehen zu müssen. Mit der Klassenvererbung von VFP brauchen Sie keinen Finger zu rühren, um Ihrem Programm diese zusätzliche Funktionalität mitzugeben.

Eine programmatische Page-Klasse?

Die Antwort auf die Frage in der Überschrift lautet „Ja, in VFP 3.0b und VFP 5.0“.

Der Builder kümmert sich um jede Seite, die Sie beim Design dem PageFrame hinzufügen. Er hilft Ihnen aber nicht, die Anzahl der Seiten zur Laufzeit zu verändern. Wenn Sie während der Laufzeit eines Programms die Anzahl der Seiten erhöhen, haben diese nicht Ihre RightClick-Methode. Obwohl wir von der Page-Klasse visuell keine Klassen ableiten können, können wir Page-Klassen via Code erzeugen. Als Beispiel könnten Sie in einer Prozedur folgendes machen:

* P_CLASS.PRG

DEFINE CLASS MyPage AS Page

  PROCEDURE RightClick

    THIS.Parent.RightClick()

  ENDPROC

ENDDEFINE

Hier ist keine Ereignissteuerung eingebaut, die auf die Änderung von PageCount reagiert, mit der Sie das Ereignis abfangen und VFPs Page-Klasse durch Ihre eigene ersetzen können. Statt PageCount zu ändern, habe ich der Klasse MyPageFrame die Methode AddPage hinzugefügt, die Sie benutzen können, um dem PageFrame weitere Seiten während der Laufzeit hinzuzufügen:

* Methode AddPage der Klasse MyPageFrame

LPARAMETERS tcPageName

SET PROC TO P_Class ADDITIVE

IF TYPE("tcPageName“) # "C“

  * der aktuelle Seitenname

  tcPageName = "Page“+ALLTRIM(STR(THIS.PageCount+1))

ENDIF

THIS.AddObject(tcPageName,"MyPage“)

* Beachten Sie, daß Seiten keine sichtbare Eigenschaft haben,

* und, anders als andere Objekte, nicht explizit 

* sichtbar gemacht zu werden brauchen, nachdem Sie

* ADDOBJECT() auf den Seiten-Container angewandt haben.

RELEASE PROCEDURE P_Class

RETURN (TYPE("THIS.“+tcPageName)) = "O“

Gegenüber dem Vorgehen, einfach PageCount zu verändern, um Seiten hinzuzufügen, hat diese Methode noch einen anderen Vorteil: Sie erlaubt es, dem neuen eingebetteten Objekt einen Namen zu geben. Die Seitennamen von VFP sind nicht sehr genau, wenn wir uns die Reihenfolge ansehen, obwohl sie uns mit der Reihenfolge der Seiten-Definition unterstützen. In Abbildung 2 des Hauptartikels habe ich drei der Überschriften unverändert gelassen, also entsprechen die Überschriften ihren Namen, so daß Sie den Unterschied zwischen der Reihenfolge der Seiten-Definition relativen Reihenfolge der Seiten, wie sie durch die PageOrder-Eigenschaft bereitgestellt wird, sehen können. Diese Reihenfolge wird in meinem Kombinationsfeld wiedergegeben. Sie können auf die Reihenfolge der Seiten jederzeit über die Pages[]-Gruppe zugreifen. So könnten Sie beschreibende Namen wie pgOptions, pgFileLoc usw. für jede Seite einsetzen. 

Versuchen Sie nicht, AddPage() in Ihrem Programm einzusetzen, bevor Sie nicht VFP 3.0 auf VFP 3.0b upgedatet haben. Ein Bug im ersten Release des Produkts hinderte eine benutzerdefinierte Pageklasse daran, ein Ereignis fehlerfrei von VFP zu empfangen. Wenn Sie AddPage() benutzten, um eine neue Seite hinzuzufügen, konnten Sie zwar die RightClick-Methode einer neuen Seite durch das Programm einfügen (MyForm.MyPageFrame1.NewPage.RightClick()), aber ein Klick mit der rechten Maustaste auf die Seite (außer auf den Tab) führte die Methode nicht aus.

In VFP 3.0 verbinden Sie alle benötigten Seiten während des Designs - unter Verwendung eines Builders, um jeder Seite den passenden RightClick-Code hinzuzufügen - zu einer Klasse MyPageFrame. Benutzen Sie dann die Methode RemoveObject() des PageFrames oder verringern Sie den Inhalt von PageCount, um nicht gebrauchte Seiten zur Laufzeit zu löschen. In VFP 3.0b und 5.0 können Sie meine Methode AddPage() benutzen, um das passende Verhalten der Seiten, die Sie zur Laufzeit hinzufügen, zu erhalten.

